
Pràctica-14:   Cares-múltiples 
Crea unes cares com aquestes, fes-li molts canvis, posa-li el nom d´algú. 

Aquest codi utilitza variables juntament amb la funció random() per generar de manera procedimental cares aleatòries. Cada 

vegada que executeu el programa, teniu una cara diferent. 

 

 
 

Programa: 

size(200, 200); 
 
//posa una cara enmig de la finestra 
float faceX = width/2; 
float faceY = height/2; 
 
// La cara més petita és la meitat de la finestra, la cara més gran és la finestra sencera 
float faceWidth = random(width/2, width); 
float faceHeight = random(height/2, height); 
 
//color de la cara 
fill(random(255), random(255), random(255)); 
 
//dibuixa el cor 
ellipse(faceX, faceY, faceWidth, faceHeight); 
 
//dimensions dels ulls 
float eyeWidth = random(faceWidth*.1, faceWidth*.25); 
float eyeHeight = random(faceHeight*.1, faceHeight*.25); 
 
//posició dels ulls 
float spaceBetweenEyes = random(eyeWidth, eyeWidth*2); 
float leftEyeX = faceX - spaceBetweenEyes/2; 
float rightEyeX = faceX + spaceBetweenEyes/2; 
float eyeY = faceY - random(faceHeight*.1, faceHeight*.25); 
 
//blanc 
fill(255); 
 
//dibuixa els ulls 
ellipse(leftEyeX, eyeY, eyeWidth, eyeHeight); 
ellipse(rightEyeX, eyeY, eyeWidth, eyeHeight); 
 
//dimensions de les pupiles 
float pupilWidth = random(eyeWidth*.1, eyeWidth*.9); 
float pupilHeight = random(eyeHeight*.1, eyeHeight*.9); 
 
//negre 
fill(0); 
 
//dibuixa les pupiles 

ellipse(leftEyeX, eyeY, pupilWidth, pupilHeight); 
ellipse(rightEyeX, eyeY, pupilWidth, pupilHeight); 
 
// mida aleatòria de la boca i Y 
float mouthWidth = random(faceWidth*.2, faceWidth*.8); 
float mouthHeight = random(faceHeight*.1, faceHeight*.3); 
float mouthY = faceY + random(faceHeight*.1, faceHeight*.25); 
 
//color de la boca 
fill(random(255), random(255), random(255)); 
 
//dibuixa de la boca 
arc(faceX, mouthY, mouthWidth, mouthHeight, 0, 3.14); 
line(faceX - mouthWidth/2, mouthY, faceX + mouthWidth/2, mouthY); 

 

 

 

 

 

 

 

 

 

Explicació: 
Aquest codi pot semblar complicat, però quan escriviu un programa com aquest, només us centreu en una petita part alhora. 

Primer, feu que aparegui un cercle de mida aleatòria. No et preocupis per res més encara! 

Només quan ho feu funcionar, penseu a pintar aquest cercle aleatòriament. 

A continuació, obteniu dos cercles més de mida aleatòria que es mostrin als ulls. 

Finalment, centra't en la boca. 

La qüestió és que no s'escriu tot aquest programa alhora. Et concentres en una petita peça alhora. 

 

 

 

 

 

 

Retoqueu les idees 
Genera aleatòriament una flor, un arbre o el teu animal preferit. 


